

EXCEL



By Bill Jelen

Better Depreciation Method Using INDEX



I was recently a judge for the ModelOff World Financial Modeling Championships in New York City. Between my judging duties, I learned a great trick from Dan Mayoh of the ModelOff question design team. The technique is useful in modeling depreciation. Most solutions use OFFSET, but that function will slow your worksheet. Dan's solution uses INDEX.

You might be familiar with the INDEX function. Normally, it returns a value from a list. For example, `=INDEX(B3:ZZ3,20)` returns the 20th item in cells B3:ZZ3. Dan's technique forces INDEX to return a cell reference instead.

Calculating Depreciation

Let's say you're creating a model of quarterly business returns for the next 20 years. Each year, you might invest in capital expenditures that need to be depreciated over a number of years. The model will contain 80 columns of numbers. Assuming straight-line depreciation and a five-year recovery period, your depreciation calculation for any quarter will be 5% of the sum of capital expenditures for the last 20 quarters.

Figure 1

	A	B	C	D	E	F
1	Period:	1Q 14	2Q 14	3Q 14	4Q 14	1Q 15
2	Period Counter:	1	2	3	4	5
3	Capital Expenditures	4995	6995	995	3995	995
4	Depreciation	0	-249.75	-599.5	-649.25	-849
5	<----- During first 20 period, sum back to B3 ----->					

Formula bar: `=-5%*SUM(OFFSET(E3,,,1,-MIN(20,E2)))`

Something unusual is happening in Excel when you type an INDEX() function on either side of a colon.

Once you get six years into the model, the depreciation calculation would be easy. For example, in cell Z4, you would take `=-5%*SUM(F3:W3)`. But this formula won't work for the first 20 periods of the model. If you tried to copy this formula back to column E, for example, it would be trying to sum the 20 columns to the left of column E, resulting in a #REF! error. (Note: In regular practice, the length of the depreciation period would likely be stored somewhere in the model. While I've simplified

the model in this article to hard-code the depreciation period for 20 quarters, a normal model would have input cells for the recovery period, so your formula would have to be able to adapt to 5, 7, or 15 years of depreciation.)

The common workaround is to use an OFFSET function, as shown in Figure 1. I could spend an entire article explaining the OFFSET function, but since I'm trying to talk you out of using it, let me shorten the explanation to say that the OFFSET formula in Figure 1 tells Excel to have the SUM function grab up to the last 20 columns but never anything to the left of column B.

Why OFFSET is Bad

The OFFSET function is amazingly flexible, but it's a drag when it comes to Excel recalculation. The Excel engine

can't predict if a formula containing OFFSET needs to be calculated or not. Usually, Excel does smart recalculation and only takes the time to calculate cells that are dependents of the cell you just changed. With OFFSET, though, the formulas must be recalculated every time.

Using INDEX

Normally, the INDEX function takes a range of cells and returns the value from one position in that range. Consider the formula fragment `INDEX($B3:E3,MAX(1,E$2-20+1))` from cell F4 in Figure 2. If you used this fragment as a standalone formula, it would return the value in cell B3 (4995). If you copied that formula to the 40th quarter, it would return the capital expenditures from the 19th period.

Yet something unusual is happening in Excel when you type an INDEX() function on either side of a colon. If you highlight the

Figure 2

		A	B	C	D	E	F	G
1	Period:		1Q 14	2Q 14	3Q 14	4Q 14	1Q 15	2Q 15
2	Period Counter:		1	2	3	4	5	6
3	Capital Expenditures		4995	6995	995	3995	995	2995
4	Depreciation		0	-249.75	-599.5	-649.25	-849	-898.75
5			←----- During first 20 period, sum back to B3 ----->					
6								

Formula bar: `=-5%*SUM(INDEX($B3:E3,MAX(1,E$2-20+1)):E3)`

INDEX fragment in cell F4 and press the F9 key to calculate that portion, it will return 4995 as expected (see Figure 3), but then the formula doesn't make sense: Excel can't apply the SUM function from 4995 to cell E3 because that isn't a range. It certainly should return an error.

Using the Evaluate Formula tool on the Formulas tab to evaluate the INDEX() portion of the formula in cell F4, you can watch as it returns a \$B\$3 cell reference instead of the value 4995. In the next step, Excel sums B3:E3 for use in the depreciation calculation in cell F4.

To further prove what's happening, open the Go To dialog box

(CTRL+G) and type `INDEX($B3:E3,MAX(1,E$2-20+1)):E3` in the Reference box. When you click OK, Excel selects the four cells in B3:E3.

Regardless of why this works, the benefit is that INDEX isn't a volatile function. Therefore, if you're working in other areas of the model, these depreciation numbers won't have to be calculated after every entry.

For information about next year's ModelOff Financial Modeling Championships, visit www.ModelOff.com. Dan Mayoh offers tips on financial modeling at www.fintega.com. **SF**

Bill Jelen is booking Excel Seminars for your chapter PD days at www.MrExcel.com. Send questions for future articles to IMA@MrExcel.com.

Figure 3

Formula bar: <code>=-5%*SUM(4995:E3)</code>
Go To dialog box: <code>SUM(number1, [number2], ...)</code>